


Προγραμματισμός με App Inventor

Να δημιουργήσετε μια εφαρμογή στο App Inventor που να «πειράζει» τις ιδιότητες μιας εικόνας. Αναλυτικά η εφαρμογή πρέπει να έχει τις παρακάτω προδιαγραφές (μέσα στις παρενθέσεις για κάθε αντικείμενο εμφανίζονται οι ιδιότητες με τις τιμές τους):

- Μια εικόνα (**Height: 200** και **Width: 200**)
- Το κουμπί “**Εμφάνισε**” εμφανίζει την εικόνα (**Enabled: False**)
- Το κουμπί “**Κρύψε**” κρύβει την εικόνα (**Enabled: True**)
- Ένα Spinner (**ElementsFromString: 50, 100, 200** και **Selection: 200**) για να αλλάζει το ύψος της εικόνας στις τιμές **50, 100, 200**
- Ένα Slider (**Width: 100, MaxValue: 200, MinValue: 50** και **ThumbPosition: 200**) για να αλλάζει το ύψος της εικόνας μεταξύ των τιμών **50** και **200**
- Ένα TextBox (**Width: 100** και **Text: 200**) για:
 - να εμφανίζει το μέγεθος της εικόνας και
 - να αλλάζει το μέγεθος της εικόνας μεταξύ των τιμών **50** και **200** (μετά από κλικ στο κουμπί “**OK**”)
- Ένα κουμπί “**OK**” για να ελέγχει την τιμή **Text** του TextBox και εφόσον είναι αποδεκτή να αλλάζει το ύψος της εικόνας σε αυτή την τιμή
- Ένα κουμπί “**<**” για να μειώνει, εφόσον η νέα τιμή είναι αποδεκτή, το ύψος της εικόνας κατά 10 pixel
- Ένα κουμπί “**>**” για να αυξάνει, εφόσον η νέα τιμή είναι αποδεκτή, το ύψος της εικόνας κατά 10 pixel
- Ένα αντικείμενο “Notifier” για να εμφανίζει μηνύματα λάθους
- Ένα αντικείμενο “AccelerometerSensor” για να επαναφέρει μετά από ανακίνηση την εφαρμογή στην αρχική της κατάσταση
- Όταν η εικόνα είναι ορατή, είναι ενεργό το κουμπί “**Κρύψε**” και ανενεργό το κουμπί “**Εμφάνισε**”
- Όταν η εικόνα είναι κρυμμένη, είναι ενεργό το κουμπί “**Εμφάνισε**” και ανενεργό το κουμπί “**Κρύψε**”
- Για τα αντικείμενα που «πειράζουν» το ύψος της εικόνα, πρέπει πάντα να συμφωνούν οι τιμές τους, δηλαδή:
 - όταν αλλάξει η επιλογή **Selection** στο Spinner πρέπει να ενημερωθεί η θέση του δείκτη **ThumbPosition** στο Slider
 - όταν αλλάξει η τιμή του **Text** στο TextBox πρέπει να ενημερωθεί η θέση του δείκτη **ThumbPosition** στο Slider
 - όταν αλλάξει η θέση του δείκτη **ThumbPosition** στο Slider πρέπει να ενημερωθεί η τιμή του **Text** στο TextBox και υπό συνθήκη η επιλογή **Selection** στο Spinner.

Σημειώσεις:


- Σχεδόν όλα τα αντικείμενα στο App Inventor ανταποκρίνονται σε ένα ή περισσότερα συμβάντα (π.χ. `when showButton .Click`)
- Δεν μπορούν να υπάρχουν δύο διαφορετικά block εντολών για το ίδιο συμβάν σε ένα αντικείμενο. Θα εμφανιστεί μήνυμα λάθους (✘) και στα δύο block (π.χ. `when showButton .Click`)
- Ένα αντικείμενο επηρεάζει ένα ή περισσότερα από τα αντικείμενα της εφαρμογής (ενδεχομένως και τον εαυτό του, όπως στα κουμπιά “**Εμφάνισε**” και “**Κρύψε**” που απενεργοποιούν τον εαυτό τους)
- Για να αλλάξει η τιμή σε μια ιδιότητα ενός αντικειμένου χρειάζεται η κατάλληλη εντολή **set** (π.χ. `set Image1 .Visible to false`)
- Σε μια εντολή **set** μπορείτε να αλλάξετε το αντικείμενο (μόνο μεταξύ όμοιων π.χ. μεταξύ των κουμπιών), την ιδιότητα και την τιμή κάνοντας κλικ στο βελάκι που εμφανίζεται δίπλα σε καθένα από αυτά
- Μπορείτε να αντιγράψετε εντολές ή και block εντολών (δεξί κλικ πάνω τους και Duplicate)
- Στο συμβάν PositionChanged για το Slider χρησιμοποιείται η εντολή **if/else if/else if** για την επίδειξη της και μόνο (στην περίπτωση αυτή δεν αποτελεί την καλύτερη προγραμματιστικά προσέγγιση). Για να συντάξετε μια **if/else if** πρέπει να κάνετε κλικ στο γρανάζι της **if** (⚙️ if) και να μεταφέρετε όσα **else if** (ή ένα **else**) χρειάζεστε από τον



χώρο αριστερά στην εντολή δεξιά ().



Εικόνα 1. Σχεδίαση Εφαρμογής



Εικόνα 2. Αντικείμενα Εφαρμογής

Blocks

```
when showButton .Click
do
  set Image1 . Visible to true
  set showButton . Enabled to false
  set hideButton . Enabled to true

when hideButton .Click
do
  set Image1 . Visible to false
  set showButton . Enabled to true
  set hideButton . Enabled to false

when imageSizeSpinner .AfterSelecting
  selection
do
  set Image1 . Height to get selection
  set imageSizeSlider . ThumbPosition to get selection
```

```

when imageSizeSlider .PositionChanged
thumbPosition
do
  initialize local tP to floor get thumbPosition
  in
    set Image1 . Height to get tP
    if
      get tP = 50
    then
      set imageSizeSpinner . Selection to get tP
    else if
      get tP = 100
    then
      set imageSizeSpinner . Selection to get tP
    else if
      get tP = 200
    then
      set imageSizeSpinner . Selection to get tP
    set imageSizeTextBox . Text to get tP
  
```

```

when okButton .Click
do
  if
    50 ≤ imageSizeTextBox . Text and imageSizeTextBox . Text ≤ 200
  then
    set Image1 . Height to imageSizeTextBox . Text
    set imageSizeSlider . ThumbPosition to imageSizeTextBox . Text
  else
    call Notifier1 . ShowMessageDialog
      message " Δεκτές τιμές μεταξύ 50 και 200! "
      title " Προσοχή "
      buttonText " OK "
  
```

```

when increaseButton .Click
do
  if
    imageSizeTextBox . Text + 10 ≤ 200
  then
    set imageSizeSlider . ThumbPosition to imageSizeTextBox . Text + 10
  else
    call Notifier1 . ShowMessageDialog
      message " Δεκτές τιμές μεταξύ 50 και 200! "
      title " Προσοχή "
      buttonText " OK "
  
```

```

when decreaseButton .Click
do
  if
    50 ≤ imageSizeTextBox . Text - 10
  then
    set imageSizeSlider . ThumbPosition to imageSizeTextBox . Text - 10
  else
    call Notifier1 . ShowMessageDialog
      message " Δεκτές τιμές μεταξύ 50 και 200! "
      title " Προσοχή "
      buttonText " OK "
  
```

```

when AccelerometerSensor1 .Shaking
do
  set Image1 . Visible to true
  set showButton . Enabled to false
  set hideButton . Enabled to true
  set imageSizeSlider . ThumbPosition to 200
  
```